
This is a listing for the program controlling XX3 AHU4 with the code lines that were 
added to schedule zones and change flow settings on conference rooms and perimeter 
zones as a function of occupancy and outdoor air temperature highlighted in gray.  
The comment lines included in the program listing describe the functions that are 
being provided. 
 
The modifications associated with reducing the minimum flow rates from the original 
design value were made by simply changing a set point in the terminal unit controllers.  
This is a data base change in the controller rather than a program code change, so it 
does not show up in this program listing. 
 
Following this program listing is the sequence of operation document that was used to 
establish the new operating mode. 

------------------------------------------------------------------------------ 
*** Begin Listing For File 004.PGM *** Lines: 1 - 32767  *** 
  1000  C              ***   XXXX XX-3, AHU-04 DDC PROGRAM   *** 
  1020  C      
  1040  C     THIS PROGRAM WAS WRITTEN BY RICK AEDER, 6/29/94, JOB # E- 
       +1773. 
  1060  C     EACH BLOCK OF PROGRAM CODE IS DESCRIBED IN COMMENT 
  1080  C     LINES JUST PRIOR TO THE PPCL CODE. CALL LANDIS AND GYR AT  
  1100  C     234-9995 IF YOU HAVE ANY QUESTIONS. 
  1120  C 
  1140  C     THE FOLLOWING IS A LISTING OF ALL THE VIRTUAL POINTS USED IN  
       +THIS 
  1160  C     AIR HANDLERS PROGRAM WITH A SHORT DESCRIPTION OF EACH POINTS  
       +FUNCTION: 
  1180  C     P.I.D. LOOP VIRTUALS ARE NOT SHOWN. 
  1200  C 
  1220  C     A4HPLO      SUPPLY FAN HI STATIC PRESSURE LOCKOUT 
  1240  C     A4LPLO      SUPPLY FAN LO STATIC PRESSURE LOCKOUT 
  1260  C     A4RPLO      RETURN FAN STATIC PRESSURE LOCKOUT 
  1280  C     A4ABON      COMMAND ON/OFF BOTH A AND B FAN SYSTEMS 
  1300  C     A4AFON      COMMAND ON/OFF A FAN ONLY 
  1305  C     A4BFON      COMMAND ON/OFF B FAN ONLY 
  1412  C     A4BSFL      A4BS INSTANTANEOUS CFM READING 
  1414  C     A4BSAF      A4BS TIME AVERAGED CFM READING 
  1416  C     A4ASFL      A4AS INSTANTANEOUS CFM READING 
  1418  C     A4ASAF      A4AS TIME AVERAGED CFM READING 
  1420  C     A4BRFL      A4BR INSTANTANEOUS CFM READING 
  1440  C     A4BRAF      A4BR TIME AVERAGED CFM READING 
  1460  C     A4ARFL      A4AR INSTANTANEOUS CFM READING 
  1480  C     A4ARAF      A4AR TIME AVERAGED CFM READING 
  1500  C     A4TMR1      ISO DAMPER PROVES OPEN TIMER 



  1520  C     A4TMR2      INITIAL FAN START RAMP-UP TIMER 
  1540  C     A4TMR3      MAX COOLING TIMER, ADJ. D.A. ST. PRESS UP 
  1560  C     A4TMR4      NOT MAX COOLING TIMER, ADJ. D.A. ST. PRESS DOWN 
  1580  C     A4TMR5      MIXING DAMPER LOCKOUT TIMER - L.O. FROM OAT 
  1600  C     A4TMR6      A4RBDS ON TIMER 
  1620  C     A4TMR7      AHU-1 FILTER ALARM TIMER 
  1640  C     A4TMR8      OIP LOCKOUT RESET TIMER 
  1660  C     AHU4ON      ALL REQUESTED FANS ARE ON WITH DAMPERS OPEN 
  1680  C     A4SPD1      SUPPLY FAN SPEED OUTPUT #1 RAMP-UP TABLE 
  1700  C     A4SPD2      SUPPLY FAN SPEED OUTPUT #2 CONTROL LOOP 
  1720  C     A4SPD3      SUPPLY FAN SPEED OUTPUT #3 FAN PRESS. HI-LIMIT 
  1740  C     A4MSPD      MINIMUM SPEED OUTPUT (INITIALLY 3.5 VOLTS)V 
  1760  C     A4DPAV      DOWNSTREAM DUCT PRESSURE AVERAGE OF FOUR FLOORS 
  1780  C     A4SPSP      DOWNSTREAM DUCT STATIC PRESSURE SETPOINT 
  1800  C     A4HLSP      FAN STATIC PRESSURE HI LIMIT SETPOINT 
  1820  C     AS4SPD      OUTPUT OF THE FAN SPEED MIN STATEMENT, SENT TO  
       +VFD'S 
  1840  C     A4ECFM      TOTAL VOLUME OF AHU SERVED EXHAUST FANS +  
       +PRESSURIZATION 
  1860  C     A4X1CF      CFM VOLUME OF EF-1 
  1880  C     A4X2CF      CFM VOLUME OF EF-2 
  1900  C     A4OAMN             
  1920  C     A4AVMN      A AND B FAN AVERAGE MINIMUM AIR FLOW 
  1940  C     A4MNV1      MINIMUM AIR VOLUME OUTPUT #1 

  1960  C     A4MNV2      MINIMUM AIR VOLUME OUTPUT #2 
  1980  C     A4MNV3      MINIMUM AIR VOLUME OUTPUT #3 
  2000  C     A4AEV1      AVERAGE MINIMUM VALUE (TIME AVERAGED) 
  2020  C     A4AEV2      MINIMUM VOLUME TOGGLE CONTROL SWITCH 
  2040  C     A4XPCF      VOLUME OF PRESSURIZATION CFM 
  2060  C     A4FDAT      AVERAGE DISCHARGE AIR TEMPERATURE 
  2080  C     A4DACP      DISCHARGE AIR CONTROL POINT 
  2100  C     A4MACP      MIXED AIR CONTROL POINT 
  2120  C     A4M1CP      MIXED AIR CONTROL POINT #1 
  2140  C     A4HVCP      HEATING VALVE CONTROL POINT 
  2160  C     A4HPLO      HOT WATER CIRC PUMP LOCKOUT TIMER 
  2180  C     A4PDPA      PENTHOUSE DIFF PRESSURE ALARM 
  2200  C     A4MAAV      AVERAGE MIXED AIR TEMPERATURE   
  2220  C     A4IDLO      ISOLATION DAMPER LOCKOUT 
  2240  C     A4MDCP      MINIMUM OUTSIDE AIR DAMPER CONTROL POINT 
  2260  C     A4RDOR      RETURN DAMPER OVERRIDE SIGNAL 
  2262  C     A4AMAF      CALCULATED CFM OF A4A MINIMUM AIR FLOW 
  2264  C     A4BMAF      CALCULATED CFM OF A4B MINIMUM AIR FLOW 
  2280  C 
  2300  C     ************************************************************** 
       + 



  2320  C     THE FIRST LINES OF DDC CODE ARE A POWER FAILURE RESTART  
       +PROGRAM 
  2340  C     THIS LINE OF CODE SCANS THE END OF THE DDC PROGRAM WHERE  
       +INST. 
  2360  C     COMMAND ALL MECHANICAL EQUIPMENT TO PREPARE FOR RESTART. 
  2380  C     ************************************************************** 
       + 
  2400  C 
E 2420  ONPWRT (12980) 
  2440  C 
E 2450  TODMOD (1,1,1,1,1,4,8) 
  2455  C 
  2460  C     THIS CODE REPEATS THE PREVIOUS INSTRUCTION FOR TEN SECONDS 
  2480  C     TO INSURE PRIMARY POWER HAS RETURNED AND STABLILIZED TO THIS  
       +MBC. 
  2500  C      
E 2520  IF (SECNDS .LT. 90) THEN GOTO 12980 
  2522  C 
  2524  C     THIS CODE COMMANDS THE DOMESTIC HOT WATER RECIRCULATION PUMP  
  2526  C     ON AND THE EAST RESTROOM EXHAUST FAN (EF-02). 
  2528  C 
E 2530  IF (EF02ID .EQ. ON ) THEN ON (EF02SS) ELSE OFF (EF02SS) 
E 2532  ON (RP11SS) 
  2534  C 
  2536  C     THIS CODE COMMANDS THE FANS START/STOP CONTROL POINT ON. IF 
  2538  C     DAY/NIGHT CONTROL IS DESIRED, DISABLE THIS LINE OF CODE. 
  2540  C 
E 2542  ON (A4ABON) 
  2612  C 
  2614  C     ************************************************************** 
       + 
  2616  C     THIS BLOCK OF CODE CONTROLS THE HOT WATER CIRC PUMPS FOR  
  2618  C     HEATING COIL CONTROL.  IT WILL COMMAND THE PUMPS ON IF THE  

  2620  C     OAT IS LT 40 DEGREES AND OFF WHEN THE OAT IS ABOVE 42  
       +DEGREES. 
  2640  C     ************************************************************** 
       + 
  2660  C 
E 2680  IF (XX3OAT .LT. 39.5 .OR. AH4AHV .LT. 8.0 .OR. AH4BHV .LT. 8.0)  
       +THEN ON (A4P4AS,A4P4BS) 
E 2700  IF (XX3OAT .GT. 42.0 .AND. AH4AHV .GT. 8.5 .AND. AH4BHV .GT. 8.5)  
       +THEN OFF (A4P4AS,A4P4BS) 
  2720  C 
  2740  C     ************************************************************** 
       + 



  2760  C     *************        ALARMS AND SAFETIES         ************* 
       + 
  2780  C     ************************************************************** 
       + 
  2800  C     ************************************************************** 
       + 
  2820  C 
  2840  C     THIS CODE DELAYS THE ALARM FROM THE AIR FILTER UNTIL THE  
       +D.P.S. 
  2860  C     HAS MAINTAINED CONTACT CLOSURE FOR 60 (ADJUSTABLE) SECONDS. 
  2880  C 
E 2900  IF (A4FALM .EQ. OFF ) THEN A4TMR7 = 0.0 
E 2920  SAMPLE (5) A4TMR7 = A4TMR7 + 5.0 
E 2940  IF (A4TMR7 .GT. 60) THEN ON (A4FLAL) ELSE OFF (A4FLAL) 
  2960  C 
  2980  C     THIS CODE RESETS THE LOCK OUT ALARM POINTS TO NONE  
  3000  C     PRIORITY 10 MINUTES AFTER ANY OF THEM ARE COMMANDED TO  
  3020  C     OPERATOR PRIORITY. 
  3040  C 
E 3050  IF (A4LTLO .EQ. @OPER .OR. A4IDLO .EQ. @OPER .OR. A4RPLO .EQ. @OPER  
       +) THEN GOTO 3070 
E 3060  IF (A4SDLO .NE. @OPER .AND. A4HPLO .NE. @OPER .AND. A4LPLO .NE. @ 
       +OPER ) THEN A4TMR8 = 0.0 
E 3065  IF (A4SDLO .EQ. @OPER .OR. A4HPLO .EQ. @OPER .OR. A4LPLO .EQ. @OPER  
       +) THEN GOTO 3080 
E 3070  IF (A4LTLO .NE. @OPER .AND. A4IDLO .NE. @OPER .AND. A4RPLO .NE. @ 
       +OPER ) THEN A4TMR8 = 0.0 
E 3080  SAMPLE (5) A4TMR8 = A4TMR8 + 5.0 
E 3100  IF (A4TMR8 .GT. 600.0) THEN ON (A4LORT) ELSE OFF (A4LORT) 
E 3120  OIP (A4LORT,"P/O/P/N/A4??LO/Y") 
  3140  C 
  3160  C     IF THE SUPPLY OR RETURN DUCT SMOKE DETECTOR ACTIVATES, THIS   
  3180  C     CODE WILL COMMAND AN IMMEDIATE SHUTDOWN OF THE AHU.  TO  
       +RESTART 
  3200  C     THE FAN, AN OPERATOR MUST COMMAND POINT "A4SDLO" OFF. 
  3220  C 
E 3240  IF (A4RFSK .EQ. ON .OR. A4SFSK .EQ. ON ) THEN ON (A4SDLO) 
E 3260  IF (A4SDLO .EQ. ON ) THEN GOTO 12980 
  3262  C 
  3264  C     IF THE SUPPLY FAN (SUCTION SIDE) EXCEEDS 5" SUCTION PRESSURE 
  3266  C     THIS CODE WILL COMMAND AN IMMEDIATE SHUTDOWN OF THE AHU. TO  
  3268  C     RESTART THE OPERATOR MUST COMMAND POINT "A4LPLO" OFF. 

  3270  C 
E 3272  IF (A4ASPA .EQ. ON .OR. A4BSPA .EQ. ON ) THEN ON (A4LPLO) 
E 3274  IF (A4LPLO .EQ. ON ) THEN GOTO 12980 



  3280  C 
  3300  C     IF THE LOW TEMP DETECTOR IS ACTIVATED THE PROGRAM WILL  
       +COMMAND 
  3320  C     AN IMMEDIATE SHUTDOWN OF THE AHU. AN OPERATOR MUST MANUALLY  
       +RESET 
  3340  C     THE LOW TEMP DETECTOR AND COMMAND POINT "A4LTLO" OFF TO  
       +RESTART  
  3360  C     THE FAN SYSTEM. 
  3380  C 
E 3400  IF (A4LTDE .EQ. ON ) THEN ON (A4LTLO) 
E 3420  IF (A4LTLO .EQ. ON ) THEN GOTO 12980 
  3440  C 
  3460  C     IF THE DISCHARGE AIR STATIC PRESSURE EXCEEDS ITS HIGH LIMIT  
       +PRESSURE 
  3480  C     THE PROGRAM WILL COMMAND AN IMMEDIATE SHUTDOWN OF THE AHU. AN  
       + 
  3500  C     OPERATOR MUST COMMAND POINT "A4HPLO"  OFF TO RESTART THE FAN. 
  3520  C 
E 3540  IF (AH4FPR .GT. 5.0) THEN ON (A4HPLO) 
E 3560  IF (A4HPLO .EQ. ON ) THEN GOTO 12980 
  3580  C 
  3600  C     IF THE RETURN AIR STATIC PRESSURE EXCEEDS -2 INCHES SUCTION  
       +PRESS. 
  3620  C     ON EITHER A OR B FAN, THE PROGRAM WILL COMMAND AN IMMEDIATE  
       +SHUTDOWN 
  3640  C     OF THE AHU.  AN OPERATOR MUST COMMAND POINT "A4RPLO" OFF TO 
  3660  C     RESTART THE FAN. 
  3680  C 
E 3700  IF (A4ARFP .GT. 2.0 .OR. A4BRFP .GT. 2.0) THEN ON (A4RPLO) 
E 3720  IF (A4RPLO .EQ. ON ) THEN GOTO 12980 
  3740  C 
  3760  C     IF THE AHU IS RUNNING NORMALLY AND ANY ISOLATION DAMPER  
       +INDICATES 
  3780  C     THE DAMPER IS CLOSING, THE AHU WILL IMMEDIATLY SHUTDOWN. AN 
  3800  C     OPERATOR MUST COMMAND POINT "A4IDLO" OFF TO RESTART THE FAN. 
  3820  C 
E 3840  IF (AS4ASS .EQ. OFF ) THEN GOTO 3880 
E 3860  IF (A4RADS .EQ. OFF .OR. A4SADS .EQ. OFF ) THEN ON (A4IDLO) 
E 3880  IF (AS4BSS .EQ. OFF ) THEN GOTO 3920 
E 3900  IF (A4RBDS .EQ. OFF .OR. A4SBDS .EQ. OFF ) THEN ON (A4IDLO) 
E 3920  IF (A4IDLO .EQ. ON ) THEN GOTO 12980 
  3940  C 
  3960  C     IF BOTH A AND B FANS ARE COMMANDED TO OPERATE AND EITHER A OR  
       +B 
  3980  C     FAN IS COMMANDED TO OPERATE INDEPENDANTLY, GOTO THE SHUTDOWN  
       +SEQ. 
  4000  C 
E 4020  IF (A4ABON .EQ. ON .AND. A4AFON .EQ. ON ) THEN GOTO 12980 
E 4040  IF (A4ABON .EQ. ON .AND. A4BFON .EQ. ON ) THEN GOTO 12980 
  4060  C 
  4080  C     *********************************************************** 

  4081  C     ****************       CALCULATIONS        **************** 
  4082  C     *********************************************************** 
  4083  C     *********************************************************** 
  4084  C 
  4085  C     THE FOLLOWING COMPRISES THE CALCULATIONS USED IN THIS PROGRAM 
  4086  C     FOR DEVICE CONTROL.  THESE CALCULATIONS ARE SCANNED ON EACH  
       +PASS,  
  4087  C     THE OUTPUT PARAMETERS ARE TRANSFERRED TO PHYSICAL OUTPUTS 
  4088  C     LATER IN THE PROGRAM. 
  4089  C 
  4090  C     THESE LINES CALCULATE THE MINIMUM AIR FLOW CFM  
  4092  C 
E 4094  A4AMAF = A4AOAF * 0.0105 
E 4096  A4BMAF = A4BOAF * 0.0105 
  4098  C 
  4100  C     THE FOLLOWING COMMANDS THE CONFERANCE/TRAINING ROOM VAV BOXES 
  4105  C     FROM HIGH, TO LOW, TO ZERO MINIMUM AIR CFM BASED ON OUTDOOR 
  4110  C     TEMPERATURE AND OCCUPANCY.  IF UNOCCUPIED, THE VAV BOX WILL  
  4115  C     CONTROL TO ZERO MINIMUM CFM.  IF OCCUPIED AND THE OSA TEMP IS 
  4120  C     GREATER THAN 78, THE MINIMUM CFM WILL CONTROL AT ITS HIGH  
  4125  C     SETPOINT.  IF OCCUPIED AND OSA TEMP IS LESS THAN 76, THE 
  4130  C     MINIMUM CFM WILL CONTROL AT ITS LOW VOLUME SETPOINT. 
  4135  C 
  4140  C     THE FIRST LINES OF CODE ASSIGN THE HIGH/LOW VALUES TO  
  4145  C     LOCAL VARIABLE POINTS TO BE USED IN THE FOLLOWING SUBROUTINE. 
  4150  C 
E 4155  $LOC3 = 560 
E 4160  $LOC4 = 335 
  4165  C 
  4170  C     THE NEXT LINE JUMPS THE PROGRAM TO THE SUBROUTINE WHERE 
  4175  C     CALCULATIONS ARE PERFORMED. 
  4180  C 
E 4185  GOSUB 4370 T4102O,T4102M,$LOC3 ,$LOC4 
  4190  C 
  4195  C     THE PROGRAM SEQUENCE REPEATS 
  4200  C 
E 4205  $LOC3 = 535 
E 4210  $LOC4 = 210 
E 4215  GOSUB 4370 T4109O,T4109M,$LOC3 ,$LOC4 
E 4220  $LOC3 = 480 
E 4225  $LOC4 = 290 
E 4230  GOSUB 4370 T4113O,T4113M,$LOC3 ,$LOC4 
E 4235  $LOC3 = 1120 
E 4240  $LOC4 = 675 
E 4245  GOSUB 4370 T4115O,T4115M,$LOC3 ,$LOC4 
E 4250  $LOC3 = 540 
E 4255  $LOC4 = 330 
E 4260  GOSUB 4370 T4117O,T4117M,$LOC3 ,$LOC4 
E 4265  GOSUB 4370 T4213O,T4213M,$LOC3 ,$LOC4 
E 4270  GOSUB 4370 T4214O,T4214M,$LOC3 ,$LOC4 
E 4275  GOSUB 4370 T4216O,T4216M,$LOC3 ,$LOC4 



E 4280  GOSUB 4370 T4313O,T4313M,$LOC3 ,$LOC4 
E 4285  GOSUB 4370 T4314O,T4314M,$LOC3 ,$LOC4 
E 4290  GOSUB 4370 T4316O,T4316M,$LOC3 ,$LOC4 

E 4295  $LOC3 = 590 
E 4300  $LOC4 = 270 
E 4302  GOSUB 4370 T4121O,T4121M,$LOC3 ,$LOC4 
E 4304  $LOC3 = 450 
E 4306  $LOC4 = 270 
E 4308  GOSUB 4370 T4127O,T4127M,$LOC3 ,$LOC4 
E 4310  $LOC3 = 350 
E 4315  $LOC4 = 210 
E 4317  GOSUB 4370 T4301O,T4301M,$LOC3 ,$LOC4 
E 4320  GOSUB 4370 T4302O,T4302M,$LOC3 ,$LOC4 
E 4325  GOSUB 4370 T4324O,T4324M,$LOC3 ,$LOC4 
E 4330  GOSUB 4370 T4325O,T4325M,$LOC3 ,$LOC4 
  4335  C 
  4350  C     THIS SUBROUTINE IS WHERE THE VALUES ARE ACTUALLY ASSIGNED 
  4355  C     TO THE VAV BOXES MINIMUM AIR VOLUME CONTROL POINT. 
  4360  C 
E 4365  GOTO 4412 
E 4370  IF ($ARG1 .EQ. OFF ) THEN $ARG2 = 0.0 ELSE GOTO 4380 
E 4375  GOTO 4395 
E 4380  IF (XX3OAT .GT. 78) THEN $ARG2 = $ARG3 ELSE GOTO 4390 
E 4385  GOTO 4395 
E 4390  IF (XX3OAT .LT. 76) THEN $ARG2 = $ARG4 
E 4395  RETURN 
  4404  C 
  4406  C     THIS CODE CONTROLS THE CONFERENCE AND TRAINING ROOMS BETWEEN  
       +DAY 
  4408  C     AND NIGHT MODES BASED ON THE OCCUPANCY SENSOR. 
  4410  C 
E 4412  IF (T4102O .EQ. ON ) THEN OFF (T4102D) ELSE ON (T4102D) 
E 4414  IF (T4109O .EQ. ON ) THEN OFF (T4109D,T4113D) ELSE ON (T4109D, 
       +T4113D) 
E 4418  IF (T4115O .EQ. ON ) THEN OFF (T4115D) ELSE ON (T4115D) 
E 4420  IF (T4117O .EQ. ON ) THEN OFF (T4117D) ELSE ON (T4117D) 
E 4440  IF (T4121O .EQ. ON ) THEN OFF (T4121D) ELSE ON (T4121D) 
E 4460  IF (T4127O .EQ. ON ) THEN OFF (T4127D) ELSE ON (T4127D) 
E 4480  IF (T4213O .EQ. ON ) THEN OFF (T4213D) ELSE ON (T4213D) 
E 4500  IF (T4214O .EQ. ON ) THEN OFF (T4214D) ELSE ON (T4214D) 
E 4520  IF (T4216O .EQ. ON ) THEN OFF (T4216D) ELSE ON (T4216D) 
E 4540  IF (T4301O .EQ. ON ) THEN OFF (T4301D) ELSE ON (T4301D) 
E 4560  IF (T4302O .EQ. ON ) THEN OFF (T4302D) ELSE ON (T4302D) 
E 4580  IF (T4313O .EQ. ON ) THEN OFF (T4313D) ELSE ON (T4213D) 
E 4600  IF (T4314O .EQ. ON ) THEN OFF (T4314D) ELSE ON (T4314D) 



E 4620  IF (T4316O .EQ. ON ) THEN OFF (T4316D) ELSE ON (T4316D) 
E 4640  IF (T4324O .EQ. ON ) THEN OFF (T4324D) ELSE ON (T4324D) 
E 4660  IF (T4325O .EQ. ON ) THEN OFF (T4325D) ELSE ON (T4325D) 
  4740  C 
  4760  C     THESE LINES OF CODE CALCULATE THE CFM AIR FLOW OF THE SUPPLY 
  4780  C     AND RETURN FANS BASED ON INLET DIFFERENTIAL PRESSURE. 
  4800  C      
E 4820  A4ASFL = 4.0050 * SQRT(A4ASFF) * 10.4 
E 4840  A4BSFL = 4.0050 * SQRT(A4BSFF) * 10.4 
E 4860  A4ARFL = 4.0050 * SQRT(A4ARFF) * 9.38 
E 4880  A4BRFL = 4.0050 * SQRT(A4BRFF) * 9.38 
  4900  C 

  4920  C     THESE LINES TIME AVERAGE THE AIR FLOW READINGS FOR STABILITY. 
  4940  C 
E 4960  TIMAVG (A4ASAF,3,4,A4ASFL) 
E 4980  TIMAVG (A4BSAF,3,4,A4BSFL) 
E 4985  TIMAVG (A4ARAF,3,4,A4ARFL) 
E 4990  TIMAVG (A4BRAF,3,4,A4BRFL) 
  4995  C      
  5000  C     THIS SECTION OF PROGRAM ESTABLISHES THE SETPOINT FOR THE  
  5005  C     RETURN FAN SPEED.  IT IS DERVIED FROM THE SUPPLY FAN CFM 
  5010  C     MINUS THE EXH. AND PRESSURIZATION CFM (CALCULATED IN MBC 6) 
  5015  C 
E 5020  IF (A4ABON .EQ. ON ) THEN GOTO 5040 
E 5025  IF (A4AFON .EQ. ON ) THEN A4RFSP = A4ASAF - A4ECFM 
E 5030  IF (A4BFON .EQ. ON ) THEN A4RFSP = A4BSAF - A4ECFM 
E 5035  GOTO 5045 
E 5040  A4RFSP = (A4ASAF + A4BSAF) / 2 - A4ECFM 
E 5045  IF (A4ABON .EQ. ON ) THEN GOTO 5065 
E 5050  IF (A4AFON .EQ. ON ) THEN A4RFCP = A4ARAF 
E 5055  IF (A4BFON .EQ. ON ) THEN A4RFCP = A4BRAF 
E 5060  GOTO 5070 
E 5065  A4RFCP = (A4ARAF + A4BRAF) / 2 
E 5070  LOOP (128,A4RFCP,A4RSPD,A4RFSP,A4RFPG,A4RFIG,A4RFDG,2,6.5,3.0,10.0, 
       +0) 
  5075  C 
  5080  C 
  5085  C     THE FOLLOWING ESTABLISHES THE AMOUNT OF EXHAUST AIR THAT 
  5090  C     WILL BE SUBTRACTED FROM THE SUPPLY FAN CFM FOR CONTROL 
  5095  C     OF THE RETURN FAN SPEED. POINT A4XPCF EQUALS THE AMOUNT 
  5096  C     OF PRESSURIZATION CFM REQUIRED FOR BUILDING STATIC, THIS 
  5097  C     VALUE IS CALCULATED IN MBC 6. 
  5100  C 
E 5102  A4XPCF = BSPCFM 
E 5105  IF (EF02SS .EQ. ON ) THEN A4X1CF = 3.8 ELSE A4X1CF = 0.0 



E 5110  A4ECFM = A4X1CF + A4XPCF 
  5202  C 
  5204  C     THIS TIMER SETS THE RAMP UP SPEED ONCE THE SUPPLY FAN STARTS. 
  5206  C 
E 5208  IF (AHU4ON .EQ. OFF ) THEN SECND2 = 0.0 
  5210  C 
E 5212  A4TMR2 = SECND2 
  5214  C 
  5216  C     THIS LINE IS THE RAMP-UP SUPPLY FAN SPEED CONTROL. 
  5218  C 
E 5220  TABLE (A4TMR2,A4SPD1,0.0,A4MSPD,600.0,10.0) 
  5240  C 
  5260  C     THIS LOOP CONTROLS THE SUPPLY FAN OPERATING SPEED BY THE  
       +AVERAGE  
  5280  C     DOWNSTREAM DUCT STATIC PRESSURE. THIS LOOP TAKES CONTROL ONCE  
       +THE  
  5300  C     RAMP-UP TIMER EXCEEDS THE OUTPUT VALUE OF THIS STATEMENT. 
  5320  C 
E 5340  A4DPAV = (A4DPR1 + A4DPR2 + A4DPR3 + A4DPR4) / 4 
E 5360  LOOP (128,A4DPAV,A4SPD2,A4SPSP,A4FSPP,A4FSPI,A4FSPD,2,6.5,3.0,10.0, 
       +0) 

  5380  C 
  5400  C     THIS LOOP WILL TAKE CONTROL OF THE FAN SPEED IF THE DISCHARGE 
  5420  C     DUCT STATIC PRESSURE EXCEEDS ITS HIGH LIMIT SETPOINT. 
  5440  C 
E 5460  LOOP (128,AH4FPR,A4SPD3,A4HLSP,A4FSPP,A4FSPI,A4FSPD,2,6.0,2.0,10.0, 
       +0) 
  5480  C 
  5500  C     THE MIN STATEMENT SELECTS THE LOWEST OUTPUT VALUE OF THE  
       +ABOVE LOOP 
  5520  C     AND TABLE STATEMENTS TO OUTPUT THE ACTUAL FAN SPEED CONTROL  
       +VALUE.  
  5540  C 
E 5560  MIN (AS4SPD,A4SPD1,A4SPD2,A4SPD3) 
  5580  C 
  5920  C     THIS CODE ALLOWS T.U. BOX 4-109 TEMPERATURE SENSOR TO CONTROL 
  5940  C     T.U. BOX 4-113 HEATING/COOLING EQUIPMENT. THESE T.U. SERVE  
       +THE 
  5960  C     SAME ROOM, ONLY ONE SENSOR SHOULD BE IN CONTROL. 
  5980  C 
E 6000  T4113I = T4109T 
E 6010  T4113O = T4109O 
  6020  C 
  6040  C     THIS SECTION OF CODE LOOKS AT ALL VAV BOXES SERVED BY THE  
       +AHU.  



  6060  C     THE RESULT IS THE BOX WITH THE HIGHEST CALL FOR COOLING. 
  6080  C 
E 6100  MAX (MX4CL1,T4101C,T4102C,T4103C,T4104C,T4105C,T4106C,T4107C,T4108C, 
       +T4109C,T4110C,T4111C,T4112C,T4113C,T4114C) 
E 6120  MAX (MX4CL2,T4115C,T4117C,T4118C,T4119C,T4120C,T4121C,T4122C,T4123C, 
       +T4124C,T4125C,T4126C,T4127C,T4201C) 
E 6140  MAX (MX4CL3,T4202C,T4203C,T4204C,T4205C,T4206C,T4207C,T4208C,T4209C, 
       +T4210C,T4211C,T4212C,T4213C,T4214C,T4215C) 
E 6160  MAX (MX4CL4,T4216C,T4217C,T4218C,T4219C,T4220C,T4221C,T4222C,T4223C, 
       +T4301C,T4302C,T4303C,T4304C,T4305C,T4306C) 
E 6180  MAX (MX4CL5,T4307C,T4308C,T4309C,T4310C,T4311C,T4312C,T4313C,T4314C, 
       +T4315C,T4316C,T4317C,T4318C,T4319C,T4320C) 
E 6200  MAX (MX4CL6,T4321C,T4322C,T4323C,T4324C,T4325C,T4401C,T4403C,T4404C, 
       +T4405C,T4406C,T4407C,T4408C,T4409C,T4410C) 
E 6220  MAX (MX4CL7,T4411C,T4412C,T4413C,T4414C,T4415C,T4416C,T4417C,T4418C, 
       +T4419C,T4420C,T4421C,T4422C,T4423C,T4424C) 
E 6240  MAX (MX4CL8,T4501C) 
E 6260  MAX (MX4CLG,MX4CL1,MX4CL2,MX4CL3,MX4CL4,MX4CL5,MX4CL6,MX4CL7,MX4CL8) 
       + 
  6280  C 
  6300  C     THIS CODE ADJUSTS THE DISCHARGE STATIC PRESSURE BASED ON  
       +DEMAND OF 
  6320  C     VAV BOX COOLING REQUIREMENTS.  ANY BOX CALLING FOR 100%  
       +COOLING 
  6340  C     WILL ADJUST THE SETPOINT UP, IF ALL BOXES CALL FOR LESS THAN  
       +100% 
  6360  C     COOLING, THE DISCHARGE AIR STATIC PRESS SETPOINT WILL ADJUST  
       +DOWN. 
  6380  C      
  6400  C     THESE LINES ADJUST THE TIMERS TO INCREASE OR DECREASE  

       +SETPOINT. 
  6420  C     A4TMR3 IS STARTED BY MX4CLG EQUAL TO 100% CALL FOR COOLING. 
  6440  C     A4TMR4 IS STARTED WHEN A4TMR3 EQUALS ZERO. 
  6460  C 
E 6465  IF (A4ABON .EQ. OFF ) THEN SET (0.75,A4SPSP) 
E 6470  IF (A4ABON .EQ. OFF ) THEN GOTO 7210 
E 6480  IF (MX4CLG .LT. 97) THEN SECND3 = 0.0 
E 6500  A4TMR3 = SECND3 
  6520  C 
E 6540  IF (A4TMR3 .GT. 3.0) THEN SECND4 = 0.0 
E 6560  A4TMR4 = SECND4 
  6580  C 
  6600  C     THESE LINES MAKE THE ADJUSTMENT UP OR DOWN TO THE DISCHARGE  
       +STATIC 
  6620  C     PRESSURE BASED ON COOLING DEMAND. STABILIZING TIME DELAYS ARE  



       +ADDED 
  6640  C     TO MAKE THE ADJUSTMENT PROCESS A GRADUAL ACTION. 
  6660  C 
  6680  C     IF THE STATIC PRESSURE SETPOINT ALREADY EXCEEDS 1.5, THEN DO  
       +NOT 
  6700  C     EXECUTE ANY ADDITIONAL UPWARD ADJUSTMENT. 
  6720  C 
E 6740  IF (A4SPSP .GE. 1.5) THEN GOTO 7000 
  6760  C 
  6780  C     ON ADDITIONAL CALL FOR COOLING, ADJ THE S.P. UP EVERY 150  
       +SECONDS 
  6800  C 
E 6820  IF (A4TMR3 .GT. 149) THEN A4SPSP = A4SPSP + 0.019 
  6840  C 
  6860  C     ONCE THE COUNTER EXCEEDS 150, RESET BACK TO ZERO AND START  
       +OVER. 
  6880  C 
E 6900  IF (A4TMR3 .GT. 149) THEN SECND3 = 0.0 
  6920  C      
  6940  C     IF THE STATIC PRESSURE SETPOINT IS LESS THAN .75, THEN DO NOT  
       + 
  6960  C     EXECUTE ANY ADDITIONAL DOWNWARD ADJUSTMENT. 
  6980  C 
E 7000  IF (A4SPSP .LE. 0.75) THEN GOTO 7210 
  7020  C 
  7040  C     IF THE CALL FOR COOLING IS SATISFIED, ADJ S.P. DOWN EA. 150  
       +SECONDS 
  7060  C 
E 7080  IF (A4TMR4 .GT. 149) THEN A4SPSP = A4SPSP - 0.019 
  7100  C 
  7120  C     ONCE THE COUNTER EXCEEDS 150 SECONDS, RESET BACK TO ZERO,  
       +START OVER 
  7140  C 
E 7160  IF (A4TMR4 .GT. 149) THEN SECND4 = 0.0 
  7180  C 
  7200  C     THE FOLLOWING CODE SETS THE MINIMUM OUTSIDE AIR VOLUME  
       +SETPOINT: 
  7206  C 
E 7210  A4XMSP = 31.0 

  7212  C 
  7240  C     IF THE SUPPLY FAN IS RUNNING AT .LT. 24 CYCLES, MIN AIR IS  
       +ZERO.  
  7260  C 
E 7280  IF (AS4AVS .LT. 4.0 .OR. AS4BVS .LT. 4.0) THEN A4AMSP = 0.0 ELSE  
       +GOTO 7320 



E 7300  GOTO 7335 
  7302  C 
  7304  C     IF OSA IS .GT. 97 DEGREES, MINIMUM AIR SETPOINT WILL BE  
  7306  C     60 PERCENT OF NORMAL VOLUME. 
  7308  C 
E 7320  IF (XX3OAT .GT. 97.0) THEN A4AMSP = A4XMSP * 0.6 
E 7330  IF (XX3OAT .LT. 90.0) THEN A4AMSP = A4XMSP 
E 7335  A4AMSP = A4XMSP 
  7340  C 
  7360  C     THIS LINE TOTALS THE MINIMUM AIR VOLUME OF A AND B FANS  
  7380  C 
E 7420  A4MNV1 = A4AMAF + A4BMAF 
  7500  C 
  7520  C     * THIS LINE TIME AVERAGES THE MINIMUM AIRFLOW FOR STABLE  
       +CONTROL 
  7540  C 
E 7560  TIMAVG (A4MNV2,3,3,A4MNV1) 
  7580  C 
  7600  C     * THIS LOOP CONTROLS THE MINIMUM AIR DAMPER POSITION PER SET  
       +POINT 
  7620  C 
E 7640  LOOP (128,A4MNV2,A4MDCP,A4AMSP,A4MAFP,A4MAFI,A4MAFD,1,8.0,2.0,14.0, 
       +0) 
  7660  C      
  7900  C 
  8060  C     THE FOLLOWING ADJUSTS THE SETPOINT OF THE EA, RA DAMPERS.  
       +THIS 
  8080  C     LINE WORKS WITH THE LOOP STATEMENT CONTROLLING THE EA, RA  
       +DAMPERS 
  8100  C     THE FINAL OUTPUT SETS THE EA, RA DAMPERS THROUGH A MAX  
       +STATEMENT 
  8120  C     LATER IN THE PROGRAM. 
  8140  C 
E 8150  A4AVMN = A4MNV2 
E 8160  LOOP (128,A4AVMN,A4RDOR,A4AMSP,A4MOPG,A4MOIG,A4MODG,2,5.0,2.0,8.0,0) 
       + 
E 8180  GOTO 8360 
E 8200  A4RDOR = 2.0 
  8220  C 
  8240  C     THE FOLLOWING CONTROLS THE DISCHARGE AIR TEMPERATURE OF AHU-4 
  8260  C     BY POSITIONING THE HTG, CLG VALVES, OA, RA, AND EA DAMPERS.  
  8280  C 
  8300  C     THIS LINE ADJUSTS THE DISCHARGE AIR TEMPERATURE BASED ON  
       +OUTDOOR  
  8320  C     TEMPERATURE. 
  8340  C 
E 8360  TABLE (XX3OAT,A4DASP,40.0,60.0,50.0,55.0) 
  8380  C 

  8400  C     THE FOLLOWING LOOP SETS THE OUTPUT PARAMETERS FOR THE  
       +DISCHARGE 
  8420  C     AIR TEMPERATURE, 3 TABLE STATEMENTS FOLLOW TO CONTROL EACH  
       +DEVICE. 
  8440  C 
E 8460  LOOP (0,A4FDAT,A4DACP,A4DASP,A4DAPG,A4DAIG,A4DADG,1,50.0,0.0,100.0, 
       +0) 
  8480  C 
  8500  C     THIS TABLE CONTROLS THE HEATING VALVES OF A AND/OR B FAN 
  8520  C 
E 8540  TABLE (A4DACP,A4HVCP,0,2.5,38,13.5) 
  8560  C 
  8580  C     THIS TABLE CREATES THE CONTROL POINT FOR THE MIXING DAMPERS 
  8600  C 
E 8620  TABLE (A4DACP,A4M1CP,39,2.5,61,13.5) 
  8640  C 
  8660  C     THIS MAX STATEMENT LOOKS AT MIXED AIR CONTROL POINT (FROM  
       +TABLE) 
  8680  C     AND MIXED AIR DAMPER OVERRIDE (FROM MIN. OUTSIDE AIR VOLUME  
       +LOOP) 
  8700  C 
E 8720  MAX (A4MACP,A4M1CP,A4RDOR) 
  8740  C 
  8760  C     THIS TABLE CONTROLS THE COOLING VALVE OF A AND B FANS, AHU-4 
  8780  C 
E 8800  TABLE (A4DACP,A4CVCP,62,4.5,100,13.5) 
  8820  C 
  8840  C     ************************************************************* 
  8860  C     *******             GLOBAL COMMANDS                   ******* 
  8880  C     ************************************************************* 
  8900  C     ************************************************************* 
  8920  C 
  8940  C     THIS IS THE DAMPER OVERRIDE CONTROL 
E 8945  IF (A4DCLO .EQ. ON ) THEN OFF (A4DOLO) 
E 8950  IF (A4DOLO .EQ. ON ) THEN OFF (A4DCLO) 
E 8955  IF (A4DCLO .EQ. ON ) THEN GOTO 9280 
E 8960  IF (A4DOLO .EQ. ON ) THEN GOTO 9320 
E 8980  IF (XX3OAT .LT. 75) THEN SECND5 = 0.0 
E 9000  A4TMR5 = SECND5 
E 9020  IF (A4TMR5 .GT. 300) THEN ON (A4REC) ELSE OFF (A4REC) 
E 9021  WAIT (15,A4REC,$LOC1 ,11) 
E 9022  IF (A4REC .EQ. OFF ) THEN OFF ($LOC1 ) 
E 9023  IF (A4REC .EQ. ON ) THEN GOTO 9280 
  9040  C 
  9060  C     THIS BLOCK OF CODE OVERRIDES CONTROL OF THE PENTHOUSE RELIEF  
  9080  C     DAMPERS IF THE PENTHOUSE SPACE STATIC PRESSURE IS GREATER  
       +THAN  
  9100  C     .5 INCHES IN RESPECT TO OUTDOOR STATIC PRESSURE. 
  9120  C 
E 9140  IF (A4PHDP .GT. 0.5) THEN ON (A4PDPA) 
E 9160  IF (A4PDPA .EQ. ON ) THEN A4FPRD = 13.0 
E 9180  IF (A4PDPA .EQ. ON ) THEN GOTO 9720 
E 9200  IF (AS4ASS .EQ. OFF .AND. AS4BSS .EQ. OFF ) THEN GOTO 9720 
E 9220  A4FPRD = A4MACP 



 
 
 
 
 
 
                          XXXX XXXX 3: INTXX3                 Page: 12 
                             PPCL Program Listing                             
Monday, December 20, 1999            LGP                              2:50 PM 
                         Listing of PPCL Program: 004 
 
------------------------------------------------------------------------------ 
  9240  C 
E 9260  GOTO 9720 
E 9280  SET (2.0,A4AEAD,A4BEAD) 
E 9281  IF ($LOC1 .EQ. ON ) THEN SET (2.0,A4AOAD,A4BOAD) 
E 9300  SET (2.0,A4FPRD) 
E 9305  RELEAS (@OPER ,A4DCLO) 
E 9310  GOTO 9720 
E 9320  SET (13.5,A4AEAD,A4AOAD,A4BEAD,A4BOAD,A4FPRD) 
E 9325  RELEAS (@OPER ,A4DOLO) 
  9560  C     ************************************************************ 
  9565  C     ************************************************************ 
  9570  C 
  9580  C     THESE LINES OF CODE DETERMINE IF THE A OR B FAN IS COMMANDED  
       +TO 
  9600  C     OPERATE AS AN INDEPENDANT FAN SYSTEM. AN OPERATOR COMMAND IS  
  9620  C     REQUIRED TO COMMAND A4AFON OR A4BFON IN BOTH THE ON/OFF STATE  
       +FOR 
  9640  C     SINGLE FAN OPERATION. 
  9660  C     IF THE A OR B FAN IS COMMANDED ON INDEPENDANTLY, THE NON- 
       +RUNNING 
  9680  C     FANS EQUIPMENT IS SHUT OFF AND COMMANDED TO FAIL SAFE  
       +POSITION. 
  9682  C 
  9685  C     ************************************************************ 
  9690  C     ************************************************************ 
  9700  C 
E 9720  IF (A4AFON .EQ. ON ) THEN GOTO 11100 
E 9740  IF (A4BFON .EQ. ON ) THEN GOTO 11980 
  9760  C 
E 9780  IF (A4ABON .EQ. OFF ) THEN GOTO 12980 
E 9800  ON (A4ARID,A4BRID) 
  9820  C 
  9840  C     WHEN BOTH SUP AND RTN DAMPERS PROVE OPEN, START THE RETURN  
       +FANS. 
  9860  C 
E 9880  IF (A4RADS .EQ. OFF .OR. A4RBDS .EQ. OFF ) THEN SECND1 = 0.0 
E 9900  IF (A4RADS .EQ. ON .AND. A4RBDS .EQ. ON ) THEN ON (AR4ASS,AR4BSS) 
E 9920  A4TMR1 = SECND1 
  9940  C 
  9960  C     CODE CONTROLS STARTUP OF BOTH A AND B SUPPLY FAN/DAMPERS. 
  9980  C     TEN SECONDS AFTER THE RETURN FAN IS STARTED, THE SUPPLY ISO 
  10000 C     DAMPERS ARE OPENED. AFTER THEY PROVE OPEN, THE SUPPLY FANS  
       +ARE 
  10020 C     STARTED IN LOW SPEED. 
  10040 C 



E 10060 IF (A4TMR1 .GT. 10.0) THEN ON (A4ASID,A4BSID) 
E 10080 IF (A4SADS .EQ. ON .AND. A4SBDS .EQ. ON ) THEN ON (AS4ASS,AS4BSS)  
       +ELSE GOTO 13000 
  10100 C 
  10120 C     ONCE THE FANS (A AND B) PROVE ON AND THE DAMPERS PROVE OPEN  
       +THE  
  10140 C     SYSTEM IS RELEASED TO AUTOMATIC CONTROL. 
  10160 C 
E 10180 IF (AS4ASS .EQ. PRFON .AND. AS4BSS .EQ. PRFON ) THEN ON (AHU4ON)  

       +ELSE OFF (AHU4ON) 
E 10200 IF (AHU4ON .EQ. OFF ) THEN GOTO 13040 
  10220 C 
  10240 C     * SUPPLY FAN (B) SPD CONTROL IS ACHIEVED BY USING THE SAME  
       +OUTPUT 
  10260 C     * POINT AS THE (A) FAN (AS4SPD)  
  10280 C 
E 10300 AS4AVS = AS4SPD 
E 10320 AS4BVS = AS4SPD 
  10340 C 
  10360 C 
E 10380 IF (AR4ASS .EQ. ON .AND. AS4AVS .LT. 4.0) THEN GOTO 10440 
E 10400 AR4AVS = A4RSPD 
E 10420 GOTO 10460 
E 10440 AR4AVS = 3.0 
E 10460 AR4BVS = AR4AVS 
  10480 C 
  10500 C     MINIMUM AIR VOLUME WILL TRACK TOGETHER. 
  10520 C 
E 10540 A4AMND = A4MDCP 
E 10560 A4BMND = A4MDCP 
  10580 C 
E 10600 A4FDAT = (A4ADAT + A4BDAT) / 2 
E 10620 AH4AHV = A4HVCP 
E 10640 AH4BHV = A4HVCP 
  10660 C 
  10680 C     A4TMR5 IS THE OAT LOCKOUT, ABOVE HI LIMIT SETPOINT, OAD WILL  
       +CLOSE. 
  10700 C 
E 10720 IF (A4REC) THEN GOTO 10880 
E 10740 A4AEAD = A4MACP 
E 10760 A4AOAD = A4M1CP 
E 10780 A4BEAD = A4MACP 
E 10800 A4BOAD = A4M1CP 
  10820 C 
  10840 C     THE FOLLOWING CONTROLS THE CLG VALVES OF BOTH FANS  



  10860 C 
E 10880 AH4ACV = A4CVCP 
E 10900 AH4BCV = A4CVCP 
  10920 C 
E 10940 GOTO 13060 
  10960 C     ************************************************************* 
  10980 C 
  11000 C     THIS CODE CONTROLS THE OPERATION OF THE SYSTEM WHEN ONLY FAN 
  11020 C     "A" IS CALLED TO OPERATE. 
  11040 C 
  11060 C     ************************************************************* 
  11080 C 
E 11100 OFF (A4BRID,A4BSID,AR4BSS,AS4BSS) 
E 11120 SET (0.0,AS4BVS,AR4BVS) 
E 11140 IF (A4BMAT .GT. 45.0) THEN SET (9.0,AH4BHV) 
E 11160 SET (2.0,A4BEAD,A4BMND,A4BOAD,AH4BCV) 
E 11180 ON (A4ARID) 
E 11200 IF (A4RADS .EQ. OFF ) THEN SECND6 = 0.0 

E 11220 A4TMR6 = SECND6 
  11240 C      
  11260 C     OPEN DAMPERS,  10 SECONDS AFTER OPENING, START SUPPLY FAN.  
  11280 C 
E 11300 IF (A4AFON .EQ. ON .AND. A4RADS .EQ. ON ) THEN ON (AR4ASS) 
E 11320 IF (A4TMR6 .GT. 10.0) THEN ON (A4ASID) 
E 11340 IF (A4SADS .EQ. ON ) THEN ON (AS4ASS) ELSE GOTO 13040 
E 11360 IF (AS4ASS .EQ. PRFON ) THEN ON (AHU4ON) ELSE OFF (AHU4ON) 
E 11380 IF (AHU4ON .EQ. OFF ) THEN GOTO 13040 
  11400 C 
E 11420 AS4AVS = AS4SPD 
  11440 C 
E 11460 IF (AR4ASS .EQ. ON .AND. AS4AVS .LT. 4.0) THEN GOTO 11560 
E 11480 AR4AVS = A4RSPD 
E 11500 GOTO 11560 
E 11520 AR4AVS = 3.0 
  11540 C 
E 11560 A4AMND = A4MDCP 
  11580 C 
E 11600 A4FDAT = A4ADAT 
  11620 C 
E 11640 AH4AHV = A4HVCP 
  11660 C 
E 11680 IF (A4REC) THEN GOTO 11780 
E 11700 A4AEAD = A4MACP 
  11720 C 
E 11740 A4AOAD = A4M1CP 



  11760 C 
E 11780 AH4ACV = A4CVCP 
  11800 C 
E 11820 GOTO 13060 
  11840 C     *********************************************************** 
  11860 C 
  11880 C     THIS CODE CONTROLS THE OPERATION OF THE SYSTEM WHEN ONLY FAN 
  11900 C     "B" IS CALLED TO OPERATE. 
  11920 C 
  11940 C     *********************************************************** 
  11960 C 
E 11980 OFF (A4AFON,A4ASID,AR4ASS,AS4ASS) 
E 12000 SET (0.0,AS4AVS,AR4AVS) 
E 12020 IF (A4AMAT .GT. 45.0) THEN SET (9.0,AH4ACV) 
E 12040 SET (2.0,A4AEAD,A4AMND,A4AOAD,AH4ACV) 
  12060 C      
  12080 C     OPEN RTN FAN B ISO DAMPER ONLY. 
  12100 C 
E 12120 ON (A4BRID) 
E 12140 IF (A4RBDS .EQ. OFF ) THEN SECND6 = 0.0 
E 12160 A4TMR6 = SECND6 
  12180 C 
E 12200 IF (A4RBDS .EQ. ON ) THEN ON (AR4BSS) 
E 12220 IF (A4TMR6 .GT. 10.0) THEN ON (A4BSID) 
E 12240 IF (A4SBDS .EQ. ON ) THEN ON (AS4BSS) ELSE GOTO 13040 
  12260 C 
E 12280 IF (AS4BSS .EQ. PRFON ) THEN ON (AHU4ON) ELSE OFF (AHU4ON) 

  12300 C 
E 12320 IF (AHU4ON .EQ. OFF ) THEN GOTO 13040 
  12340 C 
E 12360 AS4BVS = AS4SPD 
  12380 C 
E 12400 IF (AR4BSS .EQ. ON .AND. AS4BVS .LT. 4.0) THEN GOTO 12460 
E 12420 AR4BVS = A4RSPD 
E 12440 GOTO 12500 
E 12460 AR4BVS = 3.0 
  12480 C 
E 12500 A4BMND = A4MDCP 
  12520 C 
E 12540 A4FDAT = A4BDAT 
  12560 C 
E 12580 AH4BHV = A4HVCP 
  12600 C 
E 12620 IF (A4REC) THEN GOTO 12720 
E 12640 A4BEAD = A4MACP 



  12660 C 
E 12680 A4BOAD = A4M1CP 
  12700 C 
E 12720 AH4BCV = A4CVCP 
  12740 C 
  12760 C     ************************************************************** 
       + 
  12780 C 
  12800 C     THIS BLOCK OF CODE IS THE SHUTDOWN MODE FOR AHU-1.  THE FIRST  
       +LINE 
  12820 C     JUMPS OVER THE SHUTDOWN SEQUENCE DURING NORMAL "ON" OPERATION 
  12840 C 
E 12860 GOTO 13100 
  12880 C 
  12900 C     THE FOLLOWING WILL COMMAND A COMPLETE SHUTDOWN OF THE AIR  
       +HANDLER 
  12920 C     AND INDEX ALL CONTROLS INTO THEIR NORMAL "OFF" CONDITION. 
  12940 C     THE HEATING VALVE WILL CYCLE BASED ON OSA TEMP.  
  12960 C 
E 12980 OFF (AS4ASS,AR4ASS,AS4BSS,AR4BSS,A4ASID,A4BSID,A4ARID,A4BRID,AHU4ON) 
       + 
E 13000 SET (2.0,A4AEAD,A4BEAD,A4AMND,A4BMND,A4AOAD,A4BOAD,AH4ACV,AH4BCV, 
       +A4FPRD) 
E 13020 SET (0.0,AS4BVS,AS4AVS,AR4AVS,AR4BVS) 
E 13040 IF (XX3OAT .LT. 40) THEN SET (2.5,AH4AHV,AH4BHV) ELSE SET (8.5, 
       +AH4AHV,AH4BHV) 
  13050 C 
  13100 C     ************************************************************** 
       + 
  13110 C 
  13120 C     ENERGY SAVINGS SECTION TO CONTROL VAV BOXES 
  13130 C     ************************************************************** 
       + 
  13140 C 
  13150 C     SET ALL CONFERENCE TERMINAL UNITS TO DAY MODE. 
  13160 C     THIS PREVENTS ACCIDENTAL SWITCHING TO NIGHT MODE. 

  13170 C     THE FOLLOWING PROGRAM WILL OVERRIDE THE FLOW SETPOINTS 
  13180 C     BASED ON OCCUPANCY STATUS AND TEMPERATURE LIMITS. 
  13190 C 
E 13200 OFF (T4102D,T4121D,T4127D,T4115D,T4109D,T4113D,T4213D,T4214D,T4313D, 
       +T4314D,T4301D) 
E 13210 OFF (T4302D,T4324D,T4325D) 
  13220 C 
  13230 C     THE FOLLOWING SETS THE CONFERENCE ROOM TERMINAL UNIT 
  13240 C     FLOW SETPOINT TO ZERO IF THE OCCUPANCY POINT IS OFF AND  



  13250 C     THE SPACE TEMPERATURE IS NOT TOO HIGH OR TOO LOW. 
  13260 C      
  13270 C     CONFERENCE ROOM 3102 CONTROLLED BY T4102 CLASS A.      
  13280 C 
E 13290 IF (T4102O .NE. ON .AND. T4102T .LT. 78 .AND. T4102T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4102Q) 
E 13300 IF ((T4102O .OR. T4102T .GT. 78 .OR. T4102T .LT. 70) .AND. T4102Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4102Q) 
  13310 C 
  13320 C     CONFERENCE ROOM 3103 CONTROLLED BY T4121 CLASS A. 
  13330 C 
E 13340 IF (T4121O .NE. ON .AND. T4121T .LT. 78 .AND. T4121T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4121Q) 
E 13350 IF ((T4121O .OR. T4121T .GT. 78 .OR. T4121T .LT. 70) .AND. T4121Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4121Q) 
  13360 C 
  13370 C     CONFERENCE ROOM 3104 CONTROLLED BY T4127 CLASS A . 
  13380 C 
E 13390 IF (T4127O .NE. ON .AND. T4127T .LT. 78 .AND. T4127T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4127Q) 
E 13400 IF ((T4127O .OR. T4127T .GT. 78 .OR. T4127T .LT. 70) .AND. T4127Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4127Q) 
  13410 C 
  13420 C     CONFERENCE ROOM 3105 CONTROLLED BY T4115 CLASS B. 
  13430 C 
E 13440 IF (T4115O .NE. ON .AND. T4115T .LT. 78 .AND. T4115T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4115Q) 
E 13450 IF ((T4115O .OR. T4115T .GT. 78 .OR. T4115T .LT. 70) .AND. T4115Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4115Q) 
  13460 C 
  13470 C     CONFERENCE ROOM 3201 CONTROLLED BY T4213 CLASS A. 
  13480 C 
E 13490 IF (T4213O .NE. ON .AND. T4213T .LT. 78 .AND. T4213T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4213Q) 
E 13500 IF ((T4213O .OR. T4213T .GT. 78 .OR. T4213T .LT. 70) .AND. T4213Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4213Q) 
  13510 C 
  13520 C     CONFERENCE ROOM 3202 CONTROLLED BY T4214 CLASS A. 
  13530 C 
E 13540 IF (T4214O .NE. ON .AND. T4214T .LT. 78 .AND. T4214T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4214Q) 
E 13550 IF ((T4214O .OR. T4214T .GT. 78 .OR. T4214T .LT. 70) .AND. T4214Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4214Q) 
  13560 C 
  13570 C     CONFERENCE ROOM 3301 CONTROLLED BY T4313 CLASS B. 

  13580 C 
E 13590 IF (T4313O .NE. ON .AND. T4313T .LT. 78 .AND. T4313T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4313Q) 
E 13600 IF ((T4313O .OR. T4313T .GT. 78 .OR. T4313T .LT. 70) .AND. T4313Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4313Q) 
  13610 C 
  13620 C     CONFERENCE ROOM 3302 CONTROLLED BY T4314 CLASS B. 
  13630 C 
E 13640 IF (T4314O .NE. ON .AND. T4314T .LT. 78 .AND. T4314T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4314Q) 
E 13650 IF ((T4314O .OR. T4314T .GT. 78 .OR. T4314T .LT. 70) .AND. T4314Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4314Q) 
  13660 C 
  13670 C     CONFERENCE ROOM 3305 CONTROLLED BY T4301 CLASS A. 
  13680 C 
E 13690 IF (T4301O .NE. ON .AND. T4301T .LT. 78 .AND. T4301T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4301Q) 
E 13700 IF ((T4301O .OR. T4301T .GT. 78 .OR. T4301T .LT. 70) .AND. T4301Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4301Q) 
  13710 C 
  13720 C     CONFERENCE ROOM 3306 CONTROLLED BY T4302 CLASS A. 
  13730 C 
E 13740 IF (T4302O .NE. ON .AND. T4302T .LT. 78 .AND. T4302T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4302Q) 
E 13750 IF ((T4302O .OR. T4302T .GT. 78 .OR. T4302T .LT. 70) .AND. T4302Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4302Q) 
  13760 C 
  13770 C     CONFERENCE ROOM 3307 CONTROLLED BY T4324 CLASS A. 
  13780 C 
E 13790 IF (T4324O .NE. ON .AND. T4324T .LT. 78 .AND. T4324T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4324Q) 
E 13800 IF ((T4324O .OR. T4324T .GT. 78 .OR. T4324T .LT. 70) .AND. T4324Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4324Q) 
  13810 C 
  13820 C     CONFERENCE ROOM 3408 CONTROLLED BT T4325 CLASS A. 
  13830 C 
E 13840 IF (T4325O .NE. ON .AND. T4325T .LT. 78 .AND. T4325T .GT. 70) THEN  
       +SET (@OPER ,0.0,T4325Q) 
E 13850 IF ((T4325O .OR. T4325T .GT. 78 .OR. T4325T .LT. 70) .AND. T4325Q  
       +.EQ. @OPER ) THEN RELEAS (@OPER ,T4325Q) 
  13860 C 
  13870 C     THESE ARE THE CLASS-C CONFERENCE ROOM FLOW SETPOINT RESET  
       +CALCULATIONS.  
  13880 C     THEY ARE CONTROLLED THE SAME AS THE ABOVE TERMINAL UNITS BUT  
       +ARE SUBJECT 
  13890 C     TO THE OPERATOR DEFINED OCCUPANCY SCHEDULE. 
  13900 C 
  13910 C 
  13920 C     OPERATOR DEFINED OCCUPANCY SCHEDULE CONTROL, LEVEL-1 QUADRANT- 
       +B. 
  13930 C 
E 13940 IF (DAY .LT. 6 .AND. CRTIME .GT. L1QBOS .AND. CRTIME .LT. L1QBOP  
       +.OR. XX3OAT .LT. 25) THEN ON (L1QBOC) 
E 13950 IF ((DAY .GT. 5 .OR. CRTIME .LT. L1QBOS .OR. CRTIME .GT. L1QBOP)  

       +.AND. XX3OAT .GT. 27) THEN OFF (L1QBOC) 
  13960 C 
  13970 C     OPERATOR DEFINED OCCUPANCY SCHEDULE CONTROL, LEVEL-2 QUADRANT- 
       +B. 
  13980 C 
E 13990 IF (DAY .LT. 6 .AND. CRTIME .GT. L2QBOS .AND. CRTIME .LT. L2QBOP  
       +.OR. XX3OAT .LT. 25) THEN ON (L2QBOC) 
E 14000 IF ((DAY .GT. 5 .OR. CRTIME .LT. L2QBOS .OR. CRTIME .GT. L2QBOP)  
       +.AND. XX3OAT .GT. 27) THEN OFF (L2QBOC) 
  14010 C 
  14020 C     OPERATOR DEFINED OCCUPANCY SCHEDULE CONTROL, LEVEL-3 QUADRANT- 
       +B. 
  14030 C 
E 14040 IF (DAY .LT. 6 .AND. CRTIME .GT. L3QBOS .AND. CRTIME .LT. L3QBOP  
       +.OR. XX3OAT .LT. 25) THEN ON (L3QBOC) 
E 14050 IF ((DAY .GT. 5 .OR. CRTIME .LT. L3QBOS .OR. CRTIME .GT. L3QBOP)  
       +.AND. XX3OAT .GT. 27) THEN OFF (L3QBOC) 
  14060 C 
  14070 C     TRAINING ROOM 3101 TREATED AS CLASS C CONF ROOM. T4109  
       +CONTROLLING T4113 
  14080 C 
E 14090 IF (L1QBOC .NE. ON .AND. T4109O .NE. ON .AND. T4109T .LT. 78 .AND.  
       +T4109T .GT. 70) THEN SET (@OPER ,0.0,T4109Q,T4113Q) 
E 14100 IF ((L1QBOC .OR. T4109O .OR. T4109T .GT. 78 .OR. T4109T .LT. 70)  
       +.AND. T4109Q .EQ. @OPER ) THEN RELEAS (@OPER ,T4109Q,T4113Q) 
  14110 C 
  14120 C 
  14130 C     THIS IS THE LEVEL-1, QUADRANT-B TERMINAL UNIT FLOW SETPOINT  
       +OVERRIDE 
  14140 C     CALCULATIONS. FLOW SETPOINTS ARE RESET BASED ON THE OPERATOR  
       +DEFINED 
  14150 C     OCCUPANCY SCHEDULE AND SPACE TEMPERATURE LIMITS. 
  14160 C 
  14170 C 
E 14180 GOSUB 14360 L1QBOC,T4101T,T4101Q 
E 14190 GOSUB 14360 L1QBOC,T4103T,T4103Q 
E 14200 GOSUB 14360 L1QBOC,T4104T,T4104Q 
E 14210 GOSUB 14360 L1QBOC,T4105T,T4105Q 
E 14220 GOSUB 14360 L1QBOC,T4106T,T4106Q 
E 14230 GOSUB 14360 L1QBOC,T4107T,T4107Q 
E 14240 GOSUB 14360 L1QBOC,T4110T,T4110Q 
E 14250 GOSUB 14360 L1QBOC,T4114T,T4114Q 
E 14260 GOSUB 14360 L1QBOC,T4118T,T4118Q 
E 14270 GOSUB 14360 L1QBOC,T4119T,T4119Q 
E 14280 GOSUB 14360 L1QBOC,T4122T,T4122Q 
E 14290 GOSUB 14360 L1QBOC,T4123T,T4123Q 
E 14300 GOSUB 14360 L1QBOC,T4124T,T4124Q 
  14310 C 



  14320 C 
E 14330 GOTO 14410 
  14340 C 
  14350 C 
E 14360 IF ($ARG1 .NE. ON .AND. $ARG2 .LT. 78 .AND. $ARG2 .GT. 70) THEN SET  
       +(@OPER ,0.0,$ARG3 ) 

E 14370 IF (($ARG1 .EQ. ON .OR. $ARG2 .GT. 78 .OR. $ARG2 .LT. 70) .AND. $ 
       +ARG3 .EQ. @OPER ) THEN RELEAS (@OPER ,$ARG3 ) 
  14380 C 
  14390 C 
E 14400 RETURN 
  14410 C 
  14420 C 
  14430 C     THIS SECTION WILL RESET THE HEATING AND COOLING FLOW MINIMUM 
  14440 C     POINTS OF THE PERIMETER UNITS WITH REHEAT BASED ON OUTDOOR  
       +AIR 
  14450 C     TEMPERATURE. IF THE TEMPERATURE IS ABOVE THE HEATING SEASON  
  14460 C     SETPOINT, THEN THE MINIMUM FLOW SETPOINTS WILL BE SET TO  
       +ZERO. 
  14470 C     IF THE TEMPERATURE IS BELOW THE SETPOINT MINUS 3 DEGREES,  
       +THEN  
  14480 C     THE FLOW SETPOINTS WILL BE SET TO THEIR NORMAL MINIMUM VALUE. 
  14490 C     LEVEL 1 QUADRANT B 
E 14500 IF (XX3OAT .GT. JTOAES) THEN T4106M = 0 
E 14510 IF (XX3OAT .LT. JTOAES - 3) THEN T4106M = 220 
  14520 C 
E 14530 IF (XX3OAT .GT. JTOAES) THEN T4107M = 0 
E 14540 IF (XX3OAT .LT. JTOAES - 3) THEN T4107M = 115 
  14550 C 
E 14560 IF (XX3OAT .GT. JTOAES) THEN T4119M = 0 
E 14570 IF (XX3OAT .LT. JTOAES - 3) THEN T4119M = 250 
  14580 C 
E 14590 IF (XX3OAT .GT. JTOAES) THEN T4122M = 0 
E 14600 IF (XX3OAT .LT. JTOAES - 3) THEN T4122M = 250 
  14610 C 
  14620 C 
  14630 C     THIS IS THE LEVEL-2, QUADRANT-B TERMINAL UNIT FLOW SETPOINT  
       +OVERRIDE 
  14640 C     CALCULATIONS. FLOW SETPOINTS ARE RESET BASED ON THE OPERATOR  
       +DEFINED 
  14650 C     OCCUPANCY SCHEDULE AND SPACE TEMPERATURE LIMITS. 
  14660 C 
  14670 C 
E 14680 GOSUB 14910 L2QBOC,T4202T,T4202Q 
E 14690 GOSUB 14910 L2QBOC,T4203T,T4203Q 



E 14700 GOSUB 14910 L2QBOC,T4204T,T4204Q 
E 14710 GOSUB 14910 L2QBOC,T4205T,T4205Q 
E 14720 GOSUB 14910 L2QBOC,T4206T,T4206Q 
E 14730 GOSUB 14910 L2QBOC,T4207T,T4207Q 
E 14740 GOSUB 14910 L2QBOC,T4209T,T4209Q 
E 14750 GOSUB 14910 L2QBOC,T4210T,T4210Q 
E 14760 GOSUB 14910 L2QBOC,T4211T,T4211Q 
E 14770 GOSUB 14910 L2QBOC,T4212T,T4212Q 
E 14780 GOSUB 14910 L2QBOC,T4217T,T4217Q 
E 14790 GOSUB 14910 L2QBOC,T4218T,T4218Q 
E 14800 GOSUB 14910 L2QBOC,T4219T,T4219Q 
E 14810 GOSUB 14910 L2QBOC,T4220T,T4220Q 
E 14820 GOSUB 14910 L2QBOC,T4221T,T4221Q 
E 14830 GOSUB 14910 L2QBOC,T4222T,T4222Q 
E 14840 GOSUB 14910 L2QBOC,T4223T,T4223Q 

  14850 C 
  14860 C 
  14870 C 
E 14880 GOTO 14960 
  14890 C 
  14900 C 
E 14910 IF ($ARG1 .NE. ON .AND. $ARG2 .LT. 78 .AND. $ARG2 .GT. 70) THEN SET  
       +(@OPER ,0.0,$ARG3 ) 
E 14920 IF (($ARG1 .EQ. ON .OR. $ARG2 .GT. 78 .OR. $ARG2 .LT. 70) .AND. $ 
       +ARG3 .EQ. @OPER ) THEN RELEAS (@OPER ,$ARG3 ) 
  14930 C 
  14940 C 
E 14950 RETURN 
  14960 C 
  14970 C 
  14980 C     THIS SECTION WILL RESET THE HEATING AND COOLING FLOW MINIMUM 
  14990 C     POINTS OF THE PERIMETER UNITS WITH REHEAT BASED ON OUTDOOR  
       +AIR 
  15000 C     TEMPERATURE. IF THE TEMPERATURE IS ABOVE THE HEATING SEASON  
  15010 C     SETPOINT, THEN THE MINIMUM FLOW SETPOINTS WILL BE SET TO  
       +ZERO. 
  15020 C     IF THE TEMPERATURE IS BELOW THE SETPOINT MINUS 3 DEGREES,  
       +THEN  
  15030 C     THE FLOW SETPOINTS WILL BE SET TO THEIR NORMAL MINIMUM VALUE. 
  15040 C     LEVEL 2 QUADRANT B (FLOWS MATCH TEC'S) 
E 15050 IF (XX3OAT .GT. JTOAES) THEN T4206M = 0 
E 15060 IF (XX3OAT .LT. JTOAES - 3) THEN T4206M = 200 
  15070 C 
E 15080 IF (XX3OAT .GT. JTOAES) THEN T4207M = 0 
E 15090 IF (XX3OAT .LT. JTOAES - 3) THEN T4207M = 192 



  15100 C 
E 15110 IF (XX3OAT .GT. JTOAES) THEN T4217M = 0 
E 15120 IF (XX3OAT .LT. JTOAES - 3) THEN T4217M = 290 
  15130 C 
E 15140 IF (XX3OAT .GT. JTOAES) THEN T4219M = 0 
E 15150 IF (XX3OAT .LT. JTOAES - 3) THEN T4219M = 176 
  15160 C 
E 15170 IF (XX3OAT .GT. JTOAES) THEN T4220M = 0 
E 15180 IF (XX3OAT .LT. JTOAES - 3) THEN T4220M = 230 
  15190 C 
E 15200 IF (XX3OAT .GT. JTOAES) THEN T4222M = 0 
E 15210 IF (XX3OAT .LT. JTOAES - 3) THEN T4222M = 275 
  15220 C 
  15230 C 
  15240 C 
  15250 C     THIS IS THE LEVEL-3, QUADRANT-B TERMINAL UNIT FLOW SETPOINT  
       +OVERRIDE 
  15260 C     CALCULATIONS. FLOW SETPOINTS ARE RESET BASED ON THE OPERATOR  
       +DEFINED 
  15270 C     OCCUPANCY SCHEDULE AND SPACE TEMPERATURE LIMITS. 
  15280 C 
  15290 C 
E 15300 GOSUB 15520 L3QBOC,T4303T,T4303Q 
E 15310 GOSUB 15520 L3QBOC,T4304T,T4304Q 

E 15320 GOSUB 15520 L3QBOC,T4305T,T4305Q 
E 15330 GOSUB 15520 L3QBOC,T4306T,T4306Q 
E 15340 GOSUB 15520 L3QBOC,T4307T,T4307Q 
E 15350 GOSUB 15520 L3QBOC,T4309T,T4309Q 
E 15360 GOSUB 15520 L3QBOC,T4310T,T4310Q 
E 15370 GOSUB 15520 L3QBOC,T4311T,T4311Q 
E 15380 GOSUB 15520 L3QBOC,T4312T,T4312Q 
E 15390 GOSUB 15520 L3QBOC,T4317T,T4317Q 
E 15400 GOSUB 15520 L3QBOC,T4318T,T4318Q 
E 15410 GOSUB 15520 L3QBOC,T4319T,T4319Q 
E 15420 GOSUB 15520 L3QBOC,T4320T,T4320Q 
E 15430 GOSUB 15520 L3QBOC,T4321T,T4321Q 
E 15440 GOSUB 15520 L3QBOC,T4322T,T4322Q 
E 15450 GOSUB 15520 L3QBOC,T4323T,T4323Q 
  15460 C 
  15470 C 
  15480 C 
E 15490 GOTO 15570 
  15500 C 
  15510 C 
E 15520 IF ($ARG1 .NE. ON .AND. $ARG2 .LT. 78 .AND. $ARG2 .GT. 70) THEN SET  



       +(@OPER ,0.0,$ARG3 ) 
E 15530 IF (($ARG1 .EQ. ON .OR. $ARG2 .GT. 78 .OR. $ARG2 .LT. 70) .AND. $ 
       +ARG3 .EQ. @OPER ) THEN RELEAS (@OPER ,$ARG3 ) 
  15540 C 
  15550 C 
E 15560 RETURN 
  15570 C 
  15580 C 
  15590 C     THIS SECTION WILL RESET THE HEATING AND COOLING FLOW MINIMUM 
  15600 C     POINTS OF THE PERIMETER UNITS WITH REHEAT BASED ON OUTDOOR  
       +AIR 
  15610 C     TEMPERATURE. IF THE TEMPERATURE IS ABOVE THE HEATING SEASON  
  15620 C     SETPOINT, THEN THE MINIMUM FLOW SETPOINTS WILL BE SET TO  
       +ZERO. 
  15630 C     IF THE TEMPERATURE IS BELOW THE SETPOINT MINUS 3 DEGREES,  
       +THEN  
  15640 C     THE FLOW SETPOINTS WILL BE SET TO THEIR NORMAL MINIMUM VALUE. 
  15650 C     LEVEL 3 QUADRANT B (FLOWS MATCH TEC'S) 
E 15660 IF (XX3OAT .GT. JTOAES) THEN T4306M = 0 
E 15670 IF (XX3OAT .LT. JTOAES - 3) THEN T4306M = 220 
  15680 C   
E 15690 IF (XX3OAT .GT. JTOAES) THEN T4307M = 0 
E 15700 IF (XX3OAT .LT. JTOAES - 3) THEN T4307M = 200 
  15710 C 
E 15720 IF (XX3OAT .GT. JTOAES) THEN T4317M = 0 
E 15730 IF (XX3OAT .LT. JTOAES - 3) THEN T4317M = 220 
  15740 C 
E 15750 IF (XX3OAT .GT. JTOAES) THEN T4319M = 0 
E 15760 IF (XX3OAT .LT. JTOAES - 3) THEN T4319M = 170 
  15770 C 
E 15780 IF (XX3OAT .GT. JTOAES) THEN T4320M = 0 
E 15790 IF (XX3OAT .LT. JTOAES - 3) THEN T4320M = 230 
  15800 C 

E 15810 IF (XX3OAT .GT. JTOAES) THEN T4322M = 0 
E 15820 IF (XX3OAT .LT. JTOAES - 3) THEN T4322M = 230 
  15830 C 
  15840 C 
  15850 C 
E 16000 GOTO 2420 
*** End Listing For File 004.PGM *** 
  
*** End of Report *** 


